ORAU Team **NIOSH Dose Reconstruction Project** Technical Information Bulletin: Tritium Calculated and Missed Dose Estimates	Document Number: ORAUT-OTIB-0011 Effective Date: 06/29/2004 Revision No.: 00 Controlled Copy No.: _____ Page 1 of 9
Subject Expert: Scott R. Siebert Approval: Signature on File Date: 06/24/2004 Edward F. Maher, Task 5 Manager Concurrence: Signature on File Date: 06/28/2004 Richard E. Toohey, Project Director Approval: Signature on File Date: 06/29/2004 James W. Neton, OCAS Health Science Administrator	Supersedes: None

RECORD OF ISSUE/REVISIONS

ISSUE AUTHORIZATION DATE	EFFECTIVE DATE	REV. NO.	DESCRIPTION
Draft	03/17/2004	00-A	New technical information bulletin to document tritium calculated and missed dose estimates methodology. Initiated by Elizabeth Brackett.
Draft	06/17/2004	00-B	Incorporates NIOSH and internal review comments. Initiated by Elizabeth Brackett.
06/29/2004	06/29/2004	00	First approved issue. Initiated by Elizabeth Brackett.

1.0 PURPOSE

This Technical Information Bulletin provides documentation of the method for estimating tritium missed and calculated doses from urine data. To facilitate entry of organ doses into the Interactive RadioEpidemiological Program (IREP) computer code, an Excel® workbook (entitled "Tritium Doses from Urine Data Workbook.xls") was developed to create the IREP annual organ dose input data.

2.0 ALGORITHM FOR CALCULATING DOSES FROM INTAKES OF TRITIATED WATER

The model used for assessing effective dose (ED) from intakes of tritiated water are those of the International Commission on Radiological Protection (ICRP). The current model is described in detail in ICRP Publication 56 with the current dose conversion factor published in ICRP Publication 68.

Tritiated water is assumed to be completely and instantaneously absorbed into the systemic circulation whether taken in by inhalation, ingestion, or absorption through the intact skin. In a relatively short time, tritiated water equilibrates with the body water and thereafter the concentration in all body fluids, including urine, is assumed to be equal. For this reason, urine measurements of tritium are considered a direct measurement of the concentration of tritium in body water. According to ICRP Publication 23, Reference Man has a total volume of body water of 42 L distributed uniformly in soft tissue (total mass of 63 kg) and exchanges 3 L of water per day, with 1.4 L per day leaving the body via urinary excretion. Tritiated water, assuming instantaneous and uniform mixing in body water, will have a clearance rate constant k equal to 3 L per day divided by 42 L, or 7.14E-2 d^{-1}. The biological half-life is therefore 9.7 days. It is assumed that 3% of the tritiated water will be incorporated into body tissues as organically bound tritium (OBT) and will be retained with a 40-day biological half-life, determined by the exchange rate of carbon in reference man. Therefore, the metabolic model for systemic tritium is a two-compartment model with 97% incorporated as tritiated water and 3% as OBT:



Fig. 1. Tritiated water model.

Figure 1 and a detailed review of the ICRP model for tritium has been published by R. B. Richardson, et. al., Health Physics 81(3), pp. 289-301, September 2001

The ICRP, in Publication 68, reports a value of 1.8E-11 Sv/Bq as the dose conversion constant for tritiated water. This dose conversion constant accounts for the partitioning of 97% of the tritiated water to the body water and 3% to OBT. The bioassay assessment method used here relates the number of decays in the body water compartment to the effective dose. Therefore, it is necessary to correlate this dose conversion constant to the number of decays U that will occur in the body water from a 1 Bq intake I of tritiated water:

$$U = (I)(0.97)\left(\frac{1}{k}\right) = \frac{(0.97)(86400\ \frac{decays}{day})}{7.14E-2\ day^{-1}} = 1.17E+6\ decays$$

The effective dose per decay in the body water < H/U > is calculated:

$$\langle H/U \rangle = \frac{(1.8E-11\frac{Sv}{Bq})}{(1.17E+6\frac{decays}{Bq})} = 1.54E-17\frac{Sv}{decay} = 1.54E-15\frac{rem}{decay}$$

Instantaneous measurements of the concentration of tritium in urine can be used to directly estimate the effective dose from intakes of tritiated water. This is the preferred method for calculating tritium doses rather than first estimating intakes from model fits to bioassay data. The technique used here to estimate doses is based on the assumption that the area under the urine concentration "curve" is directly proportional to the number of decays that occur in the body water and, therefore, directly proportional to the effective dose. This area is estimated by the numerical integration of the individual bioassay measurements with respect to time. Depending on the length of time between measurements and value of these measurements compared to the MDA, three types of calculations are used to determine this area. With frequent urine measurements, this is accomplished with a trapezoid rule, i.e., the area is estimated with a linear function from one measurement to the next (Type 1).

This technique is modified if there is a large time gap between urine measurements because a trapezoid will over estimate the area under the curve due to elimination of tritium from the body; a 40-day time interval has been chosen as the maximum time interval for using the trapezoid technique. In this case, four different conditions need to be considered:

A. The bioassay measurements at the start and end of the gap are less than the MDA.

B. The bioassay measurement at the start of the gap is less than the MDA and the bioassay measurement at the end of the gap is greater than or equal to the MDA.

C. The bioassay measurement at the start of the gap is greater than or equal to the MDA and the measurement at the end of the gap is less than the MDA.

D. The bioassay measurements at the start and end of the gap are greater than or equal to the MDA.

It is assumed that the MDA is known for each bioassay measurement although the MDA value can change from one measurement to the next. For condition (A), the area is calculated by the product of the time gap and half the average of the two MDA values to account for missed dose. For condition (B), the area is calculated by the product of the time gap and one-half the MDA value for the

measurement at the start of the gap. Because conditions (A) and (B) are trapezoidal, they are calculated as Type 1 described above.

For conditions (C) and (D), an exponentially decreasing function that characterizes the expected decrease in urine concentration due to the 9.7-day biological clearance half-life (the ICRP rounds this to 10 days in its publications) for tritium incorporated as tritiated water is constructed starting from the measurement at the start of the gap and continuing until either the end of the gap or until the value decreases to one-half the MDA value for the measurement at the end of the gap (Type 2).

For the last bioassay measurement in the set, the same exponential function is used to estimate the area following the last point (Type 3) by extending the function to infinity. Because of the relatively short biological half-time for the clearance of tritiated water from the body, the dose calculated to infinity for this last point is assumed to have been received in the year that the urine sample was collected. The following diagram will be used to derive the proportionality constant used to convert the calculated areas to effective doses for the three method types described above:



Given: C_1 = tritium urine activity concentration (µCi/L), collected at time t_1 (days).
C_2 = tritium urine activity concentration (µCi/L), collected at time t_2 (days).
k = effective rate constant for tritiated water = 0.0714 day^{-1}.
H = effective dose equivalent (rem).

Type 1:

$$H = \left(\frac{C_1 + C_2}{2}\right)(t_2 - t_1)(42L)(1.54E-15\frac{rem}{decay})(2.22E+6\frac{dpm}{\mu Ci})(1440\frac{\min}{day}), \quad or$$

$$H = (1.03E-4)(C_1 + C_2)(t_2 - t_1) \quad \text{Equation (1)}$$

Type 2:
The area and the dose are obtained by integrating the exponential function from t_1 to t_2:

$$H = C_1\left(\frac{1-e^{-k(t_2-t_1)}}{k}\right)(42L)(1.54E-15\frac{rem}{decay})(2.22E+6\frac{dpm}{\mu Ci})(1440\frac{\min}{day}), \quad or$$

$$H = (2.89E-3)(C_1)(1-e^{-k(t_2-t_1)}) \quad \text{Equation (2)}$$

If the exponential function decreases below one-half of the MDA value during the time interval from t_1 to t_2, the exponential function is replaced with half the MDA value for the remainder of the time gap.

Type 3:
The area and the dose are obtained by integrating the exponential function starting at t_2 from 0 to ∞; the result will have the same constant as in equation (2):

$$H = (2.89E-3)(C_2) \quad \text{Equation (3)}$$

These algorithms have been implemented using Excel®. It is necessary that the bioassay dates, measurements, and MDA values exist in adjacent columns with the date column on the left, the measurement column next, and the MDA column on the right. It also is necessary for the data to be sorted in ascending order by bioassay date and time. Special care must be taken when two different bioassay measurements have the same bioassay date and time. In this case, the lower concentration value should be listed first (the earlier row). The program then will use the low value to calculate the dose in the previous interval and the higher value will be used in the subsequent interval. A zero dose will be recorded next to the first measurement because of the equal date and time.

The Tritium Doses from Urine Data Workbook has the bioassay dates (t values) in column A, the measurements (concentration values in µCi/L) in column B, and the MDA values (in µCi/L) in column C. The following formula is used to calculate doses according to equations 1-3:

Excel® Formula in Tritium Doses from Urine Data Workbook:
The dose calculation is explained using rows 2 and 3 for reference. The formula calculates the dose from the date/time in cell A2 to the date/time in cell A3. The formula in its entirety is:

```
=IF(ISBLANK(A3),(2.89*10^-3)*IF(MAX(B2,C2)=B2,B2,C2/2),IF((A3-A2)>40,IF(MAX(B2,C2)=B2,(2.89
*10^-3)*B2*(1-EXP(-0.0714*(IF((A2+LN(2*B2/C3)/0.0714)>A3,A3,(A2+LN(2*B2/C3)/0.0714))-A2)))+(1
.03*10^-4)*C3*(A3-IF((A2+LN(2*B2/C3)/0.0714)>A3,A3,(A2+LN(2*B2/C3)/0.0714))),IF(MAX(B3,C3)=
B3,(1.03*10^-4)*C2*(A3-A2),(1.03*10^-4)*(C2+C3)/2*(A3-A2))),(1.03*10^-4)*(IF(MAX(B2,C2)=B2,B2,
C2/2)+IF(MAX(B3,C3)=B3,B3,C3/2))*(A3-A2)))
```

To explain the logic of this formula, the equation will be broken out into sections, where a number in square brackets refers to another piece of the equation:

```
=IF(ISBLANK(A3),(2.89*10^-3)*IF(MAX(B2,C2)=B2,B2, C2/2),      [#1]        )
```

[#1] IF((A3-A2)>40, [#2] ,
(1.03*10^-4)*(IF(MAX(B2,C2)=B2,B2,C2/2)+IF(MAX(B3,C3)=B3,B3,C3/2))*(A3-A2))

[#2] IF(MAX(B2,C2)=B2, [#4] , [#3])

[#3] IF(MAX(B3,C3)=B3,(1.03*10^-4)*C2*(A3-A2),(1.03*10^-4)*(C2+C3)/2*(A3-A2))

[#4] (2.89*10^-3)*B2*(1-EXP(-0.0714*([#5] -A2)))+(1.03*10^-4)*C3*(A3- [#5])

[#5] IF((A2+LN(2*B2/C3)/0.0714)>A3,A3,(A2+LN(2*B2/C3)/0.0714))

These portions of the formula will be explained individually:

=IF(ISBLANK(A3),(2.89*10^-3)*IF(MAX(B2,C2)=B2,B2, C2/2), [#1])

If the date/time in the next row (row 3 in the example) is blank, then row 2 refers to the last bioassay measurement and the Type 3 equation will be used to calculate the dose. With Excel® IF statements, a true condition executes the formula after the next comma, otherwise the formula following the second comma is executed. In this case, if cell A3 is blank, the dose will be calculated by the product of 2.89E-3 and the bioassay measurement, or half the MDA if the measurement is less than the MDA. If cell A3 is not blank, the formula specified by [#1] is executed:

[#1] IF((A3-A2)>40, [#2] ,
(1.03*10^-4)*(IF(MAX(B2,C2)=B2,B2,C2/2)+IF(MAX(B3,C3)=B3,B3,C3/2))*(A3-A2))

If the time between the bioassay measurements in cells B2 and B3 exceeds 40 days, then the Type 2 equation will be used to calculate the dose using the formula referred to as [#2]. If not, the trapezoid method using equation Type 1 is used to calculate the dose. If either measurement is less than MDA, it will be replaced with half the MDA value. Formula section [#2] handles the dose calculation for a time gap greater than 40 days:

[#2] IF(MAX(B2,C2)=B2, [#4] , [#3])

If the bioassay measurement in cell B2 is less than the MDA value, formula section [#3] is used to calculate the dose:

[#3] IF(MAX(B3,C3)=B3,(1.03*10^-4)*C2*(A3-A2),(1.03*10^-4)*(C2+C3)/2*(A3-A2))

This formula section handles condition 1 and condition 2 discussed for a time gap greater than 40 days. If the bioassay measurement at the end of the gap (cell B3) also is less than the MDA value, then the trapezoid method, equation Type 1, is used to calculate the area using half the MDA values on each end of the time gap. If the bioassay measurement at the end of the gap is greater than or equal to its MDA, then half the MDA at the start of the gap is assumed to remain constant through the gap and the area is calculated from this constant value. If the measurement at the start of the gap is greater than or equal to its MDA, then the condition in formula section [#2] is true and formula section [#4] is used to calculate the dose:

[#4] (2.89*10^-3)*B2*(1-EXP(-0.0714*([#5] -A2)))+(1.03*10^-4)*C3*(A3- [#5])

The first part of this formula calculates the area using equation Type 2 to put an exponentially decreasing function from the first bioassay measurement up to the time specified by formula section [#5]. The second part of this formula calculates the area from the time specified by formula section

[#5] to the second bioassay measurement (end of the gap) using a constant value equal to half the MDA value for the second bioassay measurement. The time specified by formula section [#5] takes on one of two values:

[#5] IF((A2+LN(2*B2/C3)/0.0714)>A3,A3,(A2+LN(2*B2/C3)/0.0714))

The formula (A2+LN(2*B2/C3)/0.0714) calculates the time at which the exponentially decreasing function exactly equals half the MDA associated with the second bioassay measurement. If this time exceeds the date/time in cell A3, then the formula section [#5] is set equal to A3 and the second part of formula section [#4] is forced to be zero. This corresponds to the case where the exponentially decreasing function remains higher than half the MDA value and is used for the entire gap. If this time occurs during the gap, then the formula section [#5] is set equal to this time and the area is calculated by the exponential function (equation Type 2) up to this time and a constant value equal to half the MDA is used to calculate the remaining area.

3.0 SPECIAL CONSIDERATIONS

Special care must be taken when two different bioassay measurements have the same bioassay date and time. In this case, the lower value should be listed first (the earlier row). The program then will use the low value to calculate the dose in the previous interval and the higher value will be used in the subsequent interval. A zero dose will be recorded next to the first measurement because of the equal date and time.

4.0 DISCUSSION

4.1 Calculated vs. Missed Dose

Because both calculated dose (calculated from actual bioassay sample results) and missed dose (based on the potential missed dose due to detection limits in bioassay sampling) are calculated for the determination of probability of causation, the above-mentioned approach takes into account both positive bioassay sample results (greater than or equal to the MDA) and negative bioassay results (less than the MDA). The dose attributed to each bioassay sample result can then be determined to be either calculated (from positive sample results) or missed (from negative bioassay results).

Annual organ doses and their distribution are the input parameters for IREP. Because an individual may have left multiple samples in a single year, assigning doses as calculated and missed based on each bioassay point would be prohibitive. The simplification has been made that for any given year, the annual dose will be assumed to be calculated dose if there was a positive tritium bioassay result during that year and missed dose if all samples during that year were negative.

Calculated annual organ doses are entered in IREP with a radiation type of electron, E<15 keV, using lognormal distribution type with a geometric standard deviation (GSD) of 3. The distribution type and GSD value are the default assumptions applied by the Tritium Doses from Urine Data Workbook until uncertainty parameters associated specifically with tritium can be derived.

Missed annual organ doses are entered in IREP with a radiation type of electron, E<15 keV, using triangular distribution type. The minimum for the distribution is zero, the mode is the dose estimated as outlined in the method above, and the maximum is two times the mode value. The use of the tool to calculate the mode value for missed dose is the reason for the use of half of the MDA in all calculations in the spreadsheet. This corresponds to the maximum dose value being calculated at the MDA value.

This IREP input is formatted by the spreadsheet tool and only needs to be pasted into the appropriate IREP input file.

Use of Limiting Data Samples

The approach outlined above is based on having tritium urine samples with results available for the time period of monitoring. Whereas this is usually the case, there may be instances where an estimate is needed for a period where there are no tritium results available. For example, a missed dose calculation between the assumed start of exposure date and the first bioassay sample may be needed.

To address this type of situation, the spreadsheet tool was developed with the option of including "limiting data samples." This is accomplished by entering a date for the beginning of sampling in the spreadsheet with a result that is less than the MDA for that time period. There is an additional column in the spreadsheet tool that can be flagged for this type of limiting sample data. The tool will use the date of this limiting sample as if there was a negative sample submitted on this date and calculate the missed dose from this date to the next sample date. The column for flagging this type of limiting sample is to ensure that it is understood that there is no actual sample submitted by the individual on this date and that it is used solely for calculating the missed dose for a specified time period.

Time Periods with No Monitoring

For many individuals, there may be alternating time periods of monitoring and no monitoring for tritium. The spreadsheet tool was developed to be able to differentiate between periods where there is no monitoring when the missed dose is to be calculated and periods where there is no monitoring when no missed dose is to be calculated (assuming no exposure to tritium).

For periods where there is no monitoring and the missed dose is to be calculated, the tool will automatically calculate the missed dose between any two samples (regardless of the amount of time between samples).

For periods where there is no monitoring and the missed dose is not to be calculated (assuming no exposure to tritium), the input data to the spreadsheet tool can be entered in such a manner to accomplish this. A blank row (no data) may be entered after the final bioassay sample of a given period of monitoring and before the next period of monitoring begins. The spreadsheet tool will see the final sample of this monitoring period as effectively the final sample and will calculate the dose according to Type 3 above (calculating the integral from time 0 to ∞). The row after the blank row will then begin a new set of calculations of the tritium dose for each sample.